
Quick overview

• Based on a blog series from last year:
https://peateasea.de/
analysing-fit-data-with-perl-basic-beginnings/

• Lots of text on slides to aid non-German speakers
• Today, I’ll:

• Describe the FIT format
• Extract some data
• Plot the data
• Do a live demo! (hopefully)

FIT format

• FIT: Flexible and Interoperable Data Transfer
• Proprietary format from Garmin; now quasi-standard
• Stores data from sport, fitness, and health devices
• Designed for data to be streamed and stored in files
• A series of records containing messages:

• Definition message: describes upcoming data
• Data message: contains data fields⇒ interesting for us!

Details: https://developer.garmin.com/fit/protocol/ 2/36

Where does one get FIT files?

Some examples where FIT files are created and/or can be imported:

• Garmin, GPS navigation, sport and fitness devices
• Zwift, a virtual training app for cyclists and runners
• Rouvy, an indoor cycling app
• Strava, an online sport workout tracking
• etc.

A Zwift detour

• Zwift is a virtual training app for
cyclists and runners

• 1000s of “players” at a time
• Many courses in virtual worlds, e.g.
Watopia, London, Makuri Islands
(Japan), New York

• Connected to indoor trainer
• Controls trainer resistance, e.g. on
climbs

• Sprints, KOMs/QOMs, races
• Data in talk exported from Zwift
• I’m not affiliated with Zwift 4/36



Introducing Geo::FIT

Do we need to worry about handling data and definition messages? Hell no!

You only need Geo::FIT:

• has done all the hard work parsing FIT data for us
• provides interfaces to decode Garmin FIT files
• is based on earlier Garmin::Fit module:

• originally by Kiyokazu Suto
• extended by Matjaz Rihtar
• last release in 2017
• neither version released to CPAN

• latest release from 2024-07-13.

Using Geo::FIT

Geo::FIT processes data messages as an event stream, i.e.:

• define a callback
• register it to run on a data message type
• callback runs when FIT fetches a message from the stream, e.g. in a main loop

Later, can munge extracted data, plot, etc., as necessary.

That’s enough theory! Let’s look at a basic example.

Basic data extraction script (overview)

use strict;
use warnings;

use Geo::FIT;

my $fit = Geo::FIT->new();
$fit->file("2025-05-08-07-58-33.fit");
$fit->open or die $fit->error;

my $record_callback = sub {
my ($self, $descriptor, $values) = @_;
my $time= $self->field_value('timestamp', $descriptor, $values);
my $lat = $self->field_value('position_lat', $descriptor, $values);
my $lon = $self->field_value('position_long', $descriptor, $values);
print "Time was $time at position $lat, $lon\n";

};

$fit->data_message_callback_by_name('record', $record_callback) or die $fit->error;

my @header_things = $fit->fetch_header;

1 while ($fit->fetch);

$fit->close;

A standard opening

Basic boilerplate:

• Import modules
• Instantiate object
• Open file

use strict;
use warnings;

use Geo::FIT;

my $fit = Geo::FIT->new();
$fit->file("2025-05-08-07-58-33.fit");
$fit->open or die $fit->error;

This is a call(back)

• Callback is where the real work happens
• Run sub when see data message with name record
• Fetch field values with field_value method

# define callback
my $record_callback = sub {

my ($self, $descriptor, $values) = @_;
my $time= $self->field_value('timestamp', $descriptor, $values);
my $lat = $self->field_value('position_lat', $descriptor, $values);
my $lon = $self->field_value('position_long', $descriptor, $values);
print "Time was $time at position $lat, $lon\n";

};

# register callback
$fit->data_message_callback_by_name('record', $record_callback)

or die $fit->error; 9/36



While-ing away the time

Fetch and process each message in stream

# must fetch headers before can fetch data
my @header_things = $fit->fetch_header;

# process data
1 while ($fit->fetch);

# all done; clean up
$fit->close;

Run the code; extract the data

Let’s run the script

• we get lots of output
• there are many records in a single file
• 3273 records in this example

$ perl geo-fit-basic-data-extraction.pl
<snip>
Time was 2025-05-08T06:53:10Z at position -11.6379448 deg, 166.9560685 deg
Time was 2025-05-08T06:53:11Z at position -11.6379450 deg, 166.9560904 deg
Time was 2025-05-08T06:53:12Z at position -11.6379451 deg, 166.9561073 deg
Time was 2025-05-08T06:53:13Z at position -11.6379452 deg, 166.9561185 deg
Time was 2025-05-08T06:53:14Z at position -11.6379452 deg, 166.9561232 deg
Time was 2025-05-08T06:53:15Z at position -11.6379452 deg, 166.9561233 deg
Time was 2025-05-08T06:53:16Z at position -11.6379452 deg, 166.9561233 deg
Time was 2025-05-08T06:53:17Z at position -11.6379452 deg, 166.9561233 deg11/36



Pop quiz: where’s WallyPaulie?

Q: Where, on earth, is 11.6◦S, 166.9◦E?

Pop quiz: where’s WallyPaulie?

Q: Where, on earth, is 11.6◦S, 166.9◦E?

A: Banie Island

Pop quiz: where’s WallyPaulie?

Q: Where, on earth, is 11.6◦S, 166.9◦E?

A: Banie Island … in Temotu island group; part of Solomon Islands (blue marker)

12/36



Pop quiz: where’s WallyPaulie?

Q: Where, on earth, is 11.6◦S, 166.9◦E?

A: Banie Island … i.e., close to Australia and New Zealand

Pop quiz: where’s WallyPaulie?

Q: Where, on earth, is 11.6◦S, 166.9◦E?

A: Banie Island … not where I was, i.e., Hannover, Germany (red marker)

Is this all one can do with FIT files?

Is this all one can do with FIT files?

No! There are many fields available. Let’s find them!

Find FIT fields: callback

Again, define a callback. This time, look up the fields_list().

my $record_callback = sub {
my ($self, $descriptor, $values) = @_;
my @all_field_names = $self->fields_list($descriptor);

return \@all_field_names;
};

$fit->data_message_callback_by_name('record', $record_callback )
or die $fit->error;

Find FIT fields: main loop

This time, the main loop is more involved

• when callback isn’t called, fetch() returns 1 on success or undef
• when callback is called, fetch() returns callback value

my $found_field_names = 0;
do {

my $field_names = $fit->fetch;
my $reftype = reftype $field_names;
if (defined $reftype && $reftype eq 'ARRAY') {

print "Number of field names found: ", scalar @{$field_names}, "\n";
while (my @next_field_names = splice @{$field_names}, 0, 5) {

my $joined_field_names = join ", ", @next_field_names;
print $joined_field_names, "\n";

}
$found_field_names = 1;

}
} while (!$found_field_names); 15/36



Find FIT fields: main loop

This time, the main loop is more involved

• here, callback return value is array ref to list of field names
my $found_field_names = 0;
do {

my $field_names = $fit->fetch;
my $reftype = reftype $field_names;
if (defined $reftype && $reftype eq 'ARRAY') {

print "Number of field names found: ", scalar @{$field_names}, "\n";
while (my @next_field_names = splice @{$field_names}, 0, 5) {

my $joined_field_names = join ", ", @next_field_names;
print $joined_field_names, "\n";

}
$found_field_names = 1;

}
} while (!$found_field_names);

15/36



Find FIT fields: main loop

This time, the main loop is more involved

• only use $field_names when callback runs
• check for definedness and reftype eq 'ARRAY' (from Scalar::Util)

my $found_field_names = 0;
do {

my $field_names = $fit->fetch;
my $reftype = reftype $field_names;
if (defined $reftype && $reftype eq 'ARRAY') {

print "Number of field names found: ", scalar @{$field_names}, "\n";
while (my @next_field_names = splice @{$field_names}, 0, 5) {

my $joined_field_names = join ", ", @next_field_names;
print $joined_field_names, "\n";

}
$found_field_names = 1;

}
} while (!$found_field_names); 15/36



Find FIT fields: main loop

This time, the main loop is more involved

• there are 49 field names in this FIT file
• splice five together at a time and print

my $found_field_names = 0;
do {

my $field_names = $fit->fetch;
my $reftype = reftype $field_names;
if (defined $reftype && $reftype eq 'ARRAY') {

print "Number of field names found: ", scalar @{$field_names}, "\n";
while (my @next_field_names = splice @{$field_names}, 0, 5) {

my $joined_field_names = join ", ", @next_field_names;
print $joined_field_names, "\n";

}
$found_field_names = 1;

}
} while (!$found_field_names); 15/36



Find FIT fields: main loop

This time, the main loop is more involved

• while (splice) pattern from perldoc -f splice
• exit loop after finding field names

my $found_field_names = 0;
do {

my $field_names = $fit->fetch;
my $reftype = reftype $field_names;
if (defined $reftype && $reftype eq 'ARRAY') {

print "Number of field names found: ", scalar @{$field_names}, "\n";
while (my @next_field_names = splice @{$field_names}, 0, 5) {

my $joined_field_names = join ", ", @next_field_names;
print $joined_field_names, "\n";

}
$found_field_names = 1;

}
} while (!$found_field_names); 15/36



Find FIT fields: ~fifty fields found

Lots of information available:
Number of field names found: 49
timestamp, position_lat, position_long, distance, time_from_course
total_cycles, accumulated_power, enhanced_speed, enhanced_altitude, altitude
speed, power, grade, compressed_accumulated_power, vertical_speed
calories, vertical_oscillation, stance_time_percent, stance_time, ball_speed
cadence256, total_hemoglobin_conc, total_hemoglobin_conc_min, total_hemoglobin_conc_max, saturated_hemoglobin_percent
saturated_hemoglobin_percent_min, saturated_hemoglobin_percent_max, heart_rate, cadence, compressed_speed_distance
resistance, cycle_length, temperature, speed_1s, cycles
left_right_balance, gps_accuracy, activity_type, left_torque_effectiveness, right_torque_effectiveness
left_pedal_smoothness, right_pedal_smoothness, combined_pedal_smoothness, time128, stroke_type
zone, fractional_cadence, device_index, 1_6_target_power

Exact field list depends upon device/app

Not all fields contain data; most contain dummy values

For my use case, most interesting are:

• timestamp, position_lat, position_long, distance, altitude,
speed, power, heart_rate 16/36



Producing PNG plots in Perl

Viewing only numbers is boring. Let’s plot the data.

• extract the data from the FIT file
• plot data with Chart::Gnuplot
• gory details in blog post: https://peateasea.de/
analysing-fit-data-with-perl-producing-png-plots/

extract_activity_data(); the callback

Extract event data from fields list; avoid dummy values; store as hash1

my $record_callback = sub {
my ($self, $descriptor, $values) = @_;
my @all_field_names = $self->fields_list($descriptor);
my %event_data;
for my $field_name (@all_field_names) {

my $field_value = $self->field_value(
$field_name, $descriptor, $values);

# only real event data contains units => match on letters
if ($field_value =~ /[a-zA-Z]/) {

$event_data{$field_name} = $field_value;
}

}
return \%event_data;

};
1Full code available on GitHub: https://github.com/paultcochrane/Geo-FIT-parsing-and-plotting

extract_activity_data(); the main loop

Fetch event data with timestamp

Data without timestamp irrelevant for us

my $event_data;
my @activity_data;
do {

$event_data = $fit->fetch;
my $reftype = reftype $event_data;
if (defined $reftype && $reftype eq 'HASH'

&& defined %$event_data{'timestamp'}) {
push @activity_data, $event_data;

}
} while ($event_data);

elapsed_times(): get elapsed time data

Extract elapsed time data for x-axis
sub elapsed_times {

my $self = shift;
# get the epoch time for the first point in the time data
my @timestamps = map { $_->{'timestamp'} } $self->raw_data;
my $first_epoch_time = $date_parser

->parse_datetime($timestamps[0])->epoch;
# convert timestamp data to elapsed minutes from start of activity
my @times = map {

my $epoch_time = $date_parser->parse_datetime($_)->epoch;
my $elapsed_time = ($epoch_time - $first_epoch_time)/60;
$elapsed_time;

} @timestamps;
return @times;

}

Plot pulse and power: data setup

Pluck the data to plot from the activity data

Goal: Plot heart rate, power and altitude versus elapsed time

sub plot_activity_data {
my $self = shift;

# extract data to plot from full activity data
my @times = $self->elapsed_times;
my @heart_rates = $self->field_data_from_name('heart_rate');
my @powers = $self->field_data_from_name('power');

# ... <more code follows>
}

Plot pulse and power: chart setup

Set up chart on which to plot data
# ... plot_activity_data()
my $date = $self->date; # datestamp in plot title
my $chart = Chart::Gnuplot->new( # set up plot

output => "watopia-figure-8-heart-rate-and-power.png",
title => "Figure 8 in Watopia on $date: heart rate and power over time",
xlabel => "Elapsed time (min)",
ylabel => "Heart rate (bpm)",
terminal => "png size 1024, 768",
xtics => {incr => 5},
ytics => {mirror => "off"},
y2label => 'Power (W)',
y2range => [0, 1100],
y2tics => {incr => 100},

);

Plot pulse and power: define datasets

Define data sets for Chart::Gnuplot
# ... plot_activity_data()
# primary axis
my $heart_rate_ds = Chart::Gnuplot::DataSet->new(

xdata => \@times,
ydata => \@heart_rates,
style => "lines",

);
# secondary axis; could use e.g. altitude here
my $power_ds = Chart::Gnuplot::DataSet->new(

xdata => \@times,
ydata => \@powers,
style => "lines",
axes => "x1y2",

);

Plot pulse and power: plot in 2D

Plot the data!

{
# ... plot_activity_data()

# plot the data: power and heart rate
$chart->plot2d($power_ds, $heart_rate_ds);

# OR: plot the data, e.g. only heart rate
$chart->plot2d($heart_rate_ds);

}

That was easy, wasn’t it?

Heart rate vs elapsed time

Focus first on heart rate; ignore power/altitude data

• Why do the peaks occur where they
do? What happened?

• Let’s plot altitude as well.
• Yes, this works, even on an indoor
trainer. Gradient is translated to
trainer resistance.

Heart rate and altitude

Two peaks are hill climbs

• That explains two of the heart rate
peaks.

• What’s with the two sharp peaks?
• And what’s with the negative
altitude?

Zwifting under the deep blue sea

Negative altitude means going underground or underwater

• you can go underwater in Zwift
• e.g. through tunnels
• it’s pretty cool

Heart rate and power

The sharp peaks were from sprints!

• could run a bar heater for ∼1s
• not for much longer, though!

Compare with Zwift’s graphs

My graphs match Zwift’s

Nice to verify against another system’s output

Build your own

Where to from here?

• Don’t have to rely on a service like Zwift or Strava for graphs
• With Geo::FIT, Perl, and graphics libraries can build your own
• Can plot/compare/investigate other quantities
• Can look at long-term trends over many FIT files (e.g. detect
overtraining/undertraining; is performance improving?)

• Your data: dabble in what interests you
• Have fun!

Live demo!

What could go wrong?
The plan:
• attach Garmin head unit to computer
• see directory structure and files
• copy recent FIT file to computer
• extract FIT data in PDL session
• plot data

Demo steps

Get FIT file ready; start up PDL

# mount Garmin device
$ mount /media/edge530
# look around
$ ls /media/edge530/Garmin
$ ls /media/edge530/Garmin/Activities | tail
# copy data file to local dir; note name
$ cp <last-fit-file> .
# start pdl
$ pdl

Demo steps

pdl> use lib 'lib'; # load lib dir (why doesn't -Ilib work)
# load utils functions
pdl> use Geo::FIT::Utils
# load activity data
pdl> $activity = Geo::FIT::Utils->new(fit_file => "2025-05-08-07-58-33.fit")
# show available field names
pdl> print $activity->field_names
altitude cadence distance heart_rate position_lat position_long

power speed timestamp
# get times; load into PDL
pdl> @times = $activity->elapsed_times
pdl> $time = pdl \@times
# get heart rate data; load into PDL
pdl> @heart_rates = $activity->field_data_from_name("heart_rate")
pdl> $heart_rate = pdl \@heart_rates

Demo steps

# get speed data; load into PDL
pdl> @speeds = $activity->field_data_from_name("speed")
pdl> $speed = pdl \@speeds
# load graphics library
pdl> use PDL::Graphics::Gnuplot
# plot data on two axes
pdl> gplot(with => 'lines', $time, $speed, {axes => x1y2}, $time, $heart_rate)
# load cadence data; load into PDL
pdl> @cadences = $activity->field_data_from_name("cadence")
pdl> $cadence = pdl \@cadences
# plot all data on one graph
pdl> gplot(with => 'lines', {legend => "heart rate"}, $time, $heart_rate,

with => 'lines', {legend => "cadence"}, $time, $cadence,
with => 'lines', {axes => x1y2, legend => "speed"}, $time, $speed)

Freelance software development

Give me a yell!

Email : paul@peateasea.de
Blog : https://peateasea.de
Mastodon Mastodon: @peateasea@mastodon.social

Thanks!

